
1 Overview

 [For a Cat B CR]

This document lists all the essential changes needed to correct problems in the TTCN implementation of test case G.8.1 which is part of the iWD-TTCN3-B2016-06_D17wk10 test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Hans Rohnert

Hans.Rohnert.ext@rohde-schwarz.com
1.1 Verification Test Summary [Only for Cat B CR]
Test Case:
G.8.1
ATS Version:
ATS iWD-TTCN3-B2016-06_D17wk10
System Simulator used:
Rohde & Schwarz CMW 500 Multi-RAT Protocol Tester
UE used:
Huawei P10
Verification Status:
PASS
2 Corrections required

2.1 TC_G_8_1
	Function name
	TC_G_8_1   

	Reason for change
	In order to discover the ePDG, we need to run a DNS query  

	Summary of change
	Feed a boolean indicating usage of DNS into setup function 
Note 1: this change should be applied to all IMS over WLAN test cases. 
Note 2: using argument name “p_UseXCAP” when calling f_MTC_IMS_InitAndConnect_IP_IMS seems too narrow as we are not handling XCAP here but rather WLAN. Same for formal parameter of f_MTC_IMS_InitAndConnect_IP_IMS. 
Also, f_MTC_IMS_InitAndConnect_IP_IMS using p_UseXCAP to initialize v_HandlingOfDNS and then feeding both values into f_IP_PTC_MainLoop suggests that DNS handler and HTTP handler are independent, but they are actually traced back to the same boolean.

	TTCN module
	IMS_Testsuite_WLAN.ttcn 

	MCC160 Comment
	Accepted in principle (see below)


Before change
  testcase TC_G_8_1() runs on MTC_IMS system SYSTEM_IMS {

    // @purpose

    //   Initial registration / WLAN

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(300);

    v_IMS1 := IMS_PTC.create alive;

    f_MTC_IMS_CreateMapAndConnectPTCs_WLAN(system, v_IMS1, v_IMS2);

    v_IMS1.start(f_TC_G_8_1_IMS1());

    t_GuardTimer.start;

    f_MTC_MainLoop_IMS(t_GuardTimer);

  }
After change
  testcase TC_G_8_1() runs on MTC_IMS system SYSTEM_IMS {

    // @purpose

    //   Initial registration / WLAN

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(300);

    var boolean v_DNSuse := true;
    v_IMS1 := IMS_PTC.create alive;

    f_MTC_IMS_CreateMapAndConnectPTCs_WLAN(system, v_IMS1, v_IMS2, v_DNSuse);

    v_IMS1.start(f_TC_G_8_1_IMS1());

    t_GuardTimer.start;

    f_MTC_MainLoop_IMS(t_GuardTimer);

  }

MCC160 Implementation
	  function f_MTC_IMS_InitAndConnect_IP_IMS(SYSTEM_IMS_BASE p_System,

                                           IMS_PTC p_IMS1,

                                           IMS_PTC p_IMS2, 
                                           boolean p_UseDNS := false,
                                           boolean p_UseXCAP := false) runs on MTC_IMS

  { /* @sic R5s141051 change 3: parameter p_UseXCAP sic@ */


    var IP_PTC v_IP;

    f_MTC_BASE_Map(p_System);

    v_IP := f_IP_PTC_CreateAndMap(p_System);

    vc_Components.IMS1 := p_IMS1;

    vc_Components.IMS2 := p_IMS2;

    vc_Components.IP   := v_IP;

    connect(p_IMS1:IMS_CTRL,     v_IP:IMS_CTRL[tsc_Index_PDN1]);

    connect(p_IMS1:IMS_Server,   v_IP:IMS_Server[tsc_Index_PDN1]);

    connect(p_IMS1:IMS_Client,   v_IP:IMS_Client[tsc_Index_PDN1]);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS1], p_IMS1:MMI);

    if (p_IMS2 != null) {

      connect(p_IMS2:IMS_CTRL,   v_IP:IMS_CTRL[tsc_Index_PDN2]);

      connect(p_IMS2:IMS_Server, v_IP:IMS_Server[tsc_Index_PDN2]);

      connect(p_IMS2:IMS_Client, v_IP:IMS_Client[tsc_Index_PDN2]);

      connect(mtc:PTC_Ut[tsc_MTC_PortIndex_IMS2], p_IMS2:MMI);

      connect(p_IMS2:OtherPDN,   p_IMS1:OtherPDN);                /* @sic R5s130508 change 1 sic@ */

    }

    v_IP.start(f_IP_PTC_MainLoop(p_UseDNS, p_UseXCAP));    /* @sic R5-153795: support of ProSe - order of parameters changed sic@ */

  }

	  function f_MTC_IMS_CreateMapAndConnectPTCs_WLAN(SYSTEM_IMS p_System,

                                                  IMS_PTC p_IMS1,

                                                  IMS_PTC p_IMS2,

                                                  boolean p_UseXCAP := false) runs on MTC_IMS

  {

    var WLAN_PTC v_WLAN; 
    var boolean v_UseDNS := true;
    f_MTC_IMS_InitAndConnect_IP_IMS(p_System, p_IMS1, p_IMS2, v_UseDNS, p_UseXCAP);

    v_WLAN := WLAN_PTC.create;

    f_WLAN_PTC_Map(p_System, v_WLAN);

    connect(mtc:PTC_Ut[tsc_MTC_PortIndex_WLAN], v_WLAN:UT);

    connect(v_WLAN:IP, vc_Components.IP:WLAN_CTRL);

    if (p_IMS1 != null) { connect(vc_Components.IMS1:IPCAN, v_WLAN:IMS[tsc_Index_PDN1]); }

    if (p_IMS2 != null) { connect(vc_Components.IMS2:IPCAN, v_WLAN:IMS[tsc_Index_PDN2]); }

    f_UT_InitialPowerOffUE(Ut);

    v_WLAN.start(f_WLAN_IPCAN_MainLoop());

  }


Similar changes for 

· f_MTC_IMS_CreateMapAndConnectPTCs_EUTRA, f_MTC_IMS_XCAP_CreateMapAndConnectPTCs_EUTRA
· f_MTC_IMS_CreateMapAndConnectPTCs_UTRAN, f_MTC_IMS_XCAP_CreateMapAndConnectPTCs_UTRAN
2.2 a_DNS_Handler
	Function name
	a_DNS_Handler   

	Reason for change
	For WLAN, we do not use PDNs or DRBs. Still we want to call fl_DNS_AddressResolutionTable with an appropriate value.   

	Summary of change
	Added a branch for WLAN in which PDN index is set to 0, and skipped the DRB handling.  

	TTCN module
	Common/IP_PTC/IP_PTC_Main.ttcn 

	MCC160 Comment
	Accepted in principle (implemented as below)
Note 1: The MCC160 implmentation should work for WLAN offload and XCAP scenarios too.

Note 2: WLAN has DrbInfo.


Before change
   altstep a_DNS_Handler() runs on IP_PTC

  { /* NOTE: handling of DNS is supported for UDP only */

    /* @sic R5-145754: support of DNS sic@ */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var IP_AddrServerArray_Type v_IP_AddrServerArray;

    var AddrServerInfo_Type v_AddrServerInfo;

    var IP_DrbInfo_Type v_DrbInfo;

    var PDN_Index_Type v_PdnIndex;

    var DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable;

    [] IP_SOCK_DATA.receive(cr_UDP_DATA_IND(cr_IP_Socket(?, tsc_PortNumber_DNS_Server)))

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom;

        if (ischosen(v_LocalSocket.IpAddr.V6)) {

          v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6;

        } else {

          v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4;

        }

        v_AddrServerInfo := valueof(fl_IP_AddrServer_Get(v_IP_AddrServerArray, v_LocalSocket.IpAddr));

        v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_AddrServerInfo.PdnIndex);  /* @sic R5s150648 Change 4 sic@ */

        v_PdnIndex := v_AddrServerInfo.PdnIndex;
        v_DNS_AddressResolutionTable := fl_DNS_AddressResolutionTable(v_PdnIndex);

        v_RespData := f_DNS_GetEncodedResponse(v_Datagram.Buffer, v_DNS_AddressResolutionTable);

        if (lengthof(v_RespData) >  0) {

          fl_IP_AddrServer_SendTriggerToGeran(v_DrbInfo);               // @sic R5-133612: trigger to GERAN sic@

          IP_SOCK_DATA.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }     

After change
  altstep a_DNS_Handler() runs on IP_PTC

  { /* NOTE: handling of DNS is supported for UDP only */

    /* @sic R5-145754: support of DNS sic@ */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var IP_AddrServerArray_Type v_IP_AddrServerArray;

    var AddrServerInfo_Type v_AddrServerInfo;

    var IP_DrbInfo_Type v_DrbInfo;

    var PDN_Index_Type v_PdnIndex;

    var DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable;

    [] IP_SOCK_DATA.receive(cr_UDP_DATA_IND(cr_IP_Socket(?, tsc_PortNumber_DNS_Server)))

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom;

       if (isvalue(vc_IP_PTC_WLAN_Info.AddressInfo)) {
      

      
   v_PdnIndex:=0;

       } 
       else {
         if (ischosen(v_LocalSocket.IpAddr.V6)) {

           v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6;

         } 
         else {

           v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4;

         }

         v_PdnIndex := v_AddrServerInfo.PdnIndex;

         v_AddrServerInfo := valueof(fl_IP_AddrServer_Get(v_IP_AddrServerArray, v_LocalSocket.IpAddr));

         v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_AddrServerInfo.PdnIndex);  /* @sic R5s150648 Change 4 sic@ */

       }  


    v_DNS_AddressResolutionTable := fl_DNS_AddressResolutionTable(v_PdnIndex);

        v_RespData := f_DNS_GetEncodedResponse(v_Datagram.Buffer, v_DNS_AddressResolutionTable);

        if (lengthof(v_RespData) >  0) {

          fl_IP_AddrServer_SendTriggerToGeran(v_DrbInfo);               // @sic R5-133612: trigger to GERAN sic@

          IP_SOCK_DATA.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }     
MCC160 Implementation
	  type record of AddrServerInfo_Type IP_AddrServerArray_Type;

	  function fl_IP_DNS_Server_Get(IP_AddrInfo_Type p_ServerAddrInfo) runs on IP_PTC return template (omit) AddrServerInfo_Type

  { /* get AddrServerInfo for DNS server; in contrast to DHCP or ICMPv6 there might be a DNS server for WLAN too */

    var IP_AddrServerArray_Type v_WLAN_DNS_AddrServerArray := {};

    var IP_AddrServerArray_Type v_PDN_DNS_AddrServerArray;

    var IP_AddrServerArray_Type v_DNS_AddrServerArray;

    var boolean v_IsIPv6 := ischosen(p_ServerAddrInfo.V6);

    if (fl_WLAN_IsStarted()) { 

      if (v_IsIPv6) { v_WLAN_DNS_AddrServerArray := { vc_IP_PTC_WLAN_Info.DNS_UdpServerIPv6 }; }

      else          { v_WLAN_DNS_AddrServerArray := { vc_IP_PTC_WLAN_Info.DNS_UdpServerIPv4 }; }

    }

    if (v_IsIPv6) { v_PDN_DNS_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6; }

    else          { v_PDN_DNS_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4; }

    v_DNS_AddrServerArray := v_WLAN_DNS_AddrServerArray & v_PDN_DNS_AddrServerArray;

    return fl_IP_AddrServer_Get(v_DNS_AddrServerArray, p_ServerAddrInfo);

  }

	  function fl_WLAN_IsStarted() runs on IP_PTC return boolean

  {

    return isvalue(vc_IP_PTC_WLAN_Info.AddressInfo);

  }

	  function fl_WLAN_DrbInfo_Get() runs on IP_PTC return IP_DrbInfo_Type

  {

    return vc_IP_PTC_WLAN_Info.DrbInfo;

  }

	



















	  function fl_DNS_AddressResolutionTable_PDN(PDN_Index_Type p_PdnIndex) runs on IP_PTC return DNS_AddressResolutionTable_Type

  { /* @sic R5s160593: v_DNS_AddressResolutionTable is updated with the return value from f_DNS_AddressResolutionTable_Add sic@ */

    /* @sic R5s170448: separate AddressResolutionTables for 3GPP PDNs and WLAN sic@ */

    var template (value) DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable := {};

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution("*", v_PDN_AddressInfo.XCAP_ServerAddressIPv4, v_PDN_AddressInfo.XCAP_ServerAddressIPv6));

    return valueof(v_DNS_AddressResolutionTable);

  }

	  function fl_DNS_AddressResolutionTable_WLAN() runs on IP_PTC return DNS_AddressResolutionTable_Type

  { /* @sic R5s170448: separate AddressResolutionTables for 3GPP PDNs and WLAN sic@ */

    var template (value) DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable := {};

    var charstring v_DomainNamePatternEPDG := "epdg.epc.mnc*.mcc*.pub.3gppnetwork.org";

    var charstring v_DomainNamePatternEmergencyEPDG := "sos." & v_DomainNamePatternEPDG;

    var WLAN_AddressInfo_Type v_WLAN_AddressInfo;

    v_WLAN_AddressInfo := vc_IP_PTC_WLAN_Info.AddressInfo;

    v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEPDG,

                                                                                                                            v_WLAN_AddressInfo.EPDG_ServerAddressIPv4,

                                                                                                                            v_WLAN_AddressInfo.EPDG_ServerAddressIPv6));

    v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEmergencyEPDG,

                                                                                                                            v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv4,

                                                                                                                            v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv6));

    v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution("*", 

                                                                                                                            v_WLAN_AddressInfo.Non3GPP_XCAP_ServerAddressIPv4,

                                                                                                                            v_WLAN_AddressInfo.Non3GPP_XCAP_ServerAddressIPv6));

    return valueof(v_DNS_AddressResolutionTable);

  }

	  altstep a_DNS_Handler() runs on IP_PTC

  { /* NOTE: handling of DNS is supported for UDP only */

    /* @sic R5-145754: support of DNS sic@ */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var IP_AddrServerArray_Type v_IP_AddrServerArray;

    var AddrServerInfo_Type v_AddrServerInfo;

    var IP_DrbInfo_Type v_DrbInfo;

    var PDN_Index_Type v_PdnIndex;

    var DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable;

    [] IP_SOCK_DATA.receive(cr_UDP_DATA_IND(cr_IP_Socket(?, tsc_PortNumber_DNS_Server)))

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom;










        v_AddrServerInfo := valueof(fl_IP_DNS_Server_Get(v_LocalSocket.IpAddr));

        if (isvalue(v_AddrServerInfo.PdnIndex)) {     /* => DNS server belongs to 3GPP PDN */

          v_PdnIndex := v_AddrServerInfo.PdnIndex;

          v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_PdnIndex); /* @sic R5s150648 Change 4 sic@ */

          v_DNS_AddressResolutionTable := fl_DNS_AddressResolutionTable_PDN(v_PdnIndex);

        } else {                                      /* => DNS server belongs to WLAN (non-3GPP) */

          v_DrbInfo := fl_WLAN_DrbInfo_Get();

          v_DNS_AddressResolutionTable := fl_DNS_AddressResolutionTable_WLAN();

        }
        v_RespData := f_DNS_GetEncodedResponse(v_Datagram.Buffer, v_DNS_AddressResolutionTable);

        if (lengthof(v_RespData) >  0) {

          fl_IP_AddrServer_SendTriggerToGeran(v_DrbInfo);               // @sic R5-133612: trigger to GERAN sic@

          IP_SOCK_DATA.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }


2.3 fl_DNS_AddressResolutionTable
	Function name
	fl_DNS_AddressResolutionTable   

	Reason for change
	UEs might add a prefix before the domain name pattern for ePDG 

	Summary of change
	Added wildcard character to the pattern used.  

	TTCN module
	Common/IP_PTC/IP_PTC_Main.ttcn 

	MCC160 Comment
	Not considered as the Ue has not come UP with any DNS query for the ePDG in the provided log


Before change
  function fl_DNS_AddressResolutionTable(PDN_Index_Type p_PdnIndex) runs on IP_PTC return DNS_AddressResolutionTable_Type

  { /* @sic R5s160593: v_DNS_AddressResolutionTable is updated with the return value from f_DNS_AddressResolutionTable_Add sic@ */

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    var template (value) DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable := {};

    var charstring v_DomainNamePatternEPDG := "epdg.epc.mnc*.mcc*.pub.3gppnetwork.org";

    var charstring v_DomainNamePatternEmergencyEPDG := "sos." & v_DomainNamePatternEPDG;

    var WLAN_AddressInfo_Type v_WLAN_AddressInfo;

    if (ispresent(vc_IP_PTC_WLAN_Info.AddressInfo)) {

      v_WLAN_AddressInfo := vc_IP_PTC_WLAN_Info.AddressInfo;

      v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEPDG,

                                                                                                                              v_WLAN_AddressInfo.EPDG_ServerAddressIPv4,

                                                                                                                              v_WLAN_AddressInfo.EPDG_ServerAddressIPv6));

      v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEmergencyEPDG,

                                                                                                                              v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv4,

                                                                                                                              v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv6));

    }

    v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution("*", v_PDN_AddressInfo.XCAP_ServerAddressIPv4, v_PDN_AddressInfo.XCAP_ServerAddressIPv6));

    return valueof(v_DNS_AddressResolutionTable);

  }     

After change
   function fl_DNS_AddressResolutionTable(PDN_Index_Type p_PdnIndex) runs on IP_PTC return DNS_AddressResolutionTable_Type

  { /* @sic R5s160593: v_DNS_AddressResolutionTable is updated with the return value from f_DNS_AddressResolutionTable_Add sic@ */

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    var template (value) DNS_AddressResolutionTable_Type v_DNS_AddressResolutionTable := {};

    var charstring v_DomainNamePatternEPDG := "epdg.epc.mnc*.mcc*.pub.3gppnetwork.org";

    var charstring v_DomainNamePatternEmergencyEPDG := "sos." & v_DomainNamePatternEPDG;

    var WLAN_AddressInfo_Type v_WLAN_AddressInfo;

    var charstring v_DomainNamePatternEPDG_any := "*epdg.epc.mnc*.mcc*.pub.3gppnetwork.org"; 

    if (ispresent(vc_IP_PTC_WLAN_Info.AddressInfo)) {

      v_WLAN_AddressInfo := vc_IP_PTC_WLAN_Info.AddressInfo;

      v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEPDG_any,

                                                                                                                              v_WLAN_AddressInfo.EPDG_ServerAddressIPv4,

                                                                                                                              v_WLAN_AddressInfo.EPDG_ServerAddressIPv6));

      v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution(v_DomainNamePatternEmergencyEPDG,

                                                                                                                              v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv4,

                                                                                                                              v_WLAN_AddressInfo.EmergencyEPDG_ServerAddressIPv6));

    }

     v_DNS_AddressResolutionTable := f_DNS_AddressResolutionTable_Add(v_DNS_AddressResolutionTable, cs_DNS_AddressResolution("*", v_PDN_AddressInfo.XCAP_ServerAddressIPv4, v_PDN_AddressInfo.XCAP_ServerAddressIPv6));

    return valueof(v_DNS_AddressResolutionTable);

  }    

2.4 f_WLAN_CellConfig_Def
	Function name
	f_WLAN_CellConfig_Def   

	Reason for change
	According to ASP definition, EPDG is wlan cell non specific  

	Summary of change
	Used corresponding indication for configuration  

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn 

	MCC160 Comment
	Accepted as temporary solution: proposed implmentation is not according to ASP definitions which specify the CellId of the confirmation to be "ignored by TTCN" 
( depending on SS implementation the test case may fail
Additional change by MCC160:

UE addresses have to be distinguished for 3GPP and non-3GPP network:

The ePDG assigns the UE addresses of the 3GPP network to the UE whereas the WLAN AP assigns the non-3GPP addresses ( the ePDG needs to be configured with the PDN UE IP addresses instead of px_WLAN_UE_AddrIPv4/v6.


Before change
  function f_WLAN_CellConfig_Def(WLAN_CellId_Type p_CellId,

                                 PDN_Index_Type p_PdnIndex := PDN_Internet) runs on WLAN_PTC

  {

    var template (value) WLAN_AddressInfo_Type v_WLAN_AddressInfo := cs_WLAN_AddressInfo(px_WLAN_UE_AddrIPv4,      /* UE_IPAddressIPv4 */

                                                                                         px_WLAN_UE_AddrIPv6,      /* UE_IPAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* DNS_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* DNS_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* EPDG_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EmergencyEPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6); /* EmergencyEPDG_ServerAddressIPv6 */

    var template (value) WLAN_PDN_AddressInfo_Type v_WLAN_PDN_AddressInfo := fl_WLAN_PDN_IPAddressInfo(p_PdnIndex, valueof(v_WLAN_AddressInfo));

    var template (value) WLAN_EAP_AKA_Cntrl_Type v_EAP_AKA := cs_WLAN_EAP_AKA;

    var WLAN_AP_Parameters_Type v_AP_Parameters := f_WLAN_CellInfo_GetAP_Parameters(p_CellId);

    var WLAN_InitialCellPower_Type v_CellPower := f_WLAN_CellInfo_GetCellPower(p_CellId);

    f_IP_Handling_StartWLAN(IP, v_WLAN_AddressInfo, cs_DrbInfo_WLAN(enum2int(p_CellId)));

    f_WLAN_SS_CommonCellConfig(p_CellId, cas_WLAN_CellConfig_REQ(p_CellId, v_AP_Parameters, v_CellPower));

    f_EPDG_SS_CommonConfig(p_CellId, cas_EPDG_Config_REQ(p_CellId,

                                                         v_WLAN_PDN_AddressInfo,

                                                         cs_IPSec_SA(px_ePDG_Ipsec_IntegrityAlgorithm, px_ePDG_Ipsec_CipheringAlgorithm),

                                                         v_EAP_AKA));
  }
After change
   function f_WLAN_CellConfig_Def(WLAN_CellId_Type p_CellId,

                                 PDN_Index_Type p_PdnIndex := PDN_Internet) runs on WLAN_PTC

  {

    var template (value) WLAN_AddressInfo_Type v_WLAN_AddressInfo := cs_WLAN_AddressInfo(px_WLAN_UE_AddrIPv4,      /* UE_IPAddressIPv4 */

                                                                                         px_WLAN_UE_AddrIPv6,      /* UE_IPAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* DNS_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* DNS_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* EPDG_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EmergencyEPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6); /* EmergencyEPDG_ServerAddressIPv6 */

    var template (value) WLAN_PDN_AddressInfo_Type v_WLAN_PDN_AddressInfo := fl_WLAN_PDN_IPAddressInfo(p_PdnIndex, valueof(v_WLAN_AddressInfo));

    var template (value) WLAN_EAP_AKA_Cntrl_Type v_EAP_AKA := cs_WLAN_EAP_AKA;

    var WLAN_AP_Parameters_Type v_AP_Parameters := f_WLAN_CellInfo_GetAP_Parameters(p_CellId);

    var WLAN_InitialCellPower_Type v_CellPower := f_WLAN_CellInfo_GetCellPower(p_CellId);

    f_IP_Handling_StartWLAN(IP, v_WLAN_AddressInfo, cs_DrbInfo_WLAN(enum2int(p_CellId)));

    f_WLAN_SS_CommonCellConfig(p_CellId, cas_WLAN_CellConfig_REQ(p_CellId, v_AP_Parameters, v_CellPower));

    f_EPDG_SS_CommonConfig(wlan_Cell_NonSpecific, cas_EPDG_Config_REQ(wlan_Cell_NonSpecific,
                                                         v_WLAN_PDN_AddressInfo,

                                                         cs_IPSec_SA(px_ePDG_Ipsec_IntegrityAlgorithm, px_ePDG_Ipsec_CipheringAlgorithm),

                                                         v_EAP_AKA));
  }    

Additional MCC160 Change
	  function fl_WLAN_PDN_IPAddressInfo(PDN_Index_Type p_PdnIndex,

                                     WLAN_AddressInfo_Type p_WLAN_AddressInfo) return template(value) WLAN_PDN_AddressInfo_Type

  {

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    return cs_WLAN_PDN_AddressInfo( v_PDN_AddressInfo.UE_IPAddressIPv4,

                                    v_PDN_AddressInfo.UE_IPAddressIPv6,

                                   p_WLAN_AddressInfo.EPDG_ServerAddressIPv4,

                                   p_WLAN_AddressInfo.EPDG_ServerAddressIPv6,

                                   v_PDN_AddressInfo.PCSCF_IPAddressIPv4,

                                   v_PDN_AddressInfo.PCSCF_IPAddressIPv6);

  }


2.5 f_WLAN_CellAttenuationList
	Function name
	f_WLAN_CellAttenuationList 

	Reason for change
	f_WLAN_CellAttenuationList waits for wrong kind of confirmation  

	Summary of change
	Used proper receive template  

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn 

	MCC160 Comment
	Accepted


Before change
  function f_WLAN_CellAttenuationList(template (value) WLAN_CellPowerList_Type p_CellPowerList,

                                      template (value) WLAN_TimingInfo_Type p_TimingInfo := cs_WLAN_TimingInfoNow) runs on WLAN_PTC

  {

    var integer i;

    var WLAN_CellPower_Type v_CellPower;

    var WLAN_CellAttenuationList_Type v_CellAttenuationList;

    for (i:=0; i<lengthof(p_CellPowerList); i:=i+1) {

      v_CellPower := valueof(p_CellPowerList[i]);   // valueof cannot be avoided here

      v_CellAttenuationList[i].CellId := v_CellPower.CellId;

      v_CellAttenuationList[i].Attenuation := fl_WLAN_ChangeCellAttenuation(v_CellPower.CellId, v_CellPower.BeaconRSSI);

    }

    SYS.send(cas_WLAN_CellAtenuationList_REQ(v_CellAttenuationList, p_TimingInfo));

    SYS.receive(car_WLAN_SYSTEM_CTRL_CNF(?, cr_WLAN_CellCnf));

  }
After change
  function f_WLAN_CellAttenuationList(template (value) WLAN_CellPowerList_Type p_CellPowerList,

                                      template (value) WLAN_TimingInfo_Type p_TimingInfo := cs_WLAN_TimingInfoNow) runs on WLAN_PTC

  {

    var integer i;

    var WLAN_CellPower_Type v_CellPower;

    var WLAN_CellAttenuationList_Type v_CellAttenuationList;

    for (i:=0; i<lengthof(p_CellPowerList); i:=i+1) {

      v_CellPower := valueof(p_CellPowerList[i]);   // valueof cannot be avoided here

      v_CellAttenuationList[i].CellId := v_CellPower.CellId;

      v_CellAttenuationList[i].Attenuation := fl_WLAN_ChangeCellAttenuation(v_CellPower.CellId, v_CellPower.BeaconRSSI);

    }

    SYS.send(cas_WLAN_CellAtenuationList_REQ(v_CellAttenuationList, p_TimingInfo));

    SYS.receive(car_WLAN_SYSTEM_CTRL_CNF(?, cr_WLAN_CellAttenuationCnf)); 

  }    

2.6 f_WLAN_TurnCellOn
	Function name
	f_WLAN_TurnCellOn   

	Reason for change
	-86 dbm is not sufficient for UEs to detect WLAN cell.  

	Summary of change
	Increased power level to -50 dbm
Note: this change might merit discussion in the next SS vendors meeting, and, pending such discussion, might also result in a prose change  

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn 

	MCC160 Comment
	Accepted conditional to prose CR agreement at RAN5#76. 
NOTE: proposed solution is SS or test environment dependent


Before change
  function f_WLAN_TurnCellOn(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    f_WLAN_SetCellPower(p_CellId, f_WLAN_ConvertdBm2BeaconRSSI(-86));

  }   

After change
  function f_WLAN_TurnCellOn(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    f_WLAN_SetCellPower(p_CellId, f_WLAN_ConvertdBm2BeaconRSSI(-50));

  }    

2.7 f_WLAN_EstablishIPsecTunnel
	Function name
	f_WLAN_EstablishIPsecTunnel   

	Reason for change
	EPDG is wlan cell non specific  

	Summary of change
	For establishment of IPsec tunnel indicated wlan cell non specific property.  

	TTCN module
	Common/WLAN/WLAN_CommonProcedures.ttcn 

	MCC160 Comment
	Accepted as temporary solution (see 2.4).


Before change
    function f_WLAN_EstablishIPsecTunnel(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    timer t_Watchdog := tsc_WLANWatchDogDef;

    t_Watchdog.start;

    SYS_IND.send(cas_WLAN_SYSTEM_CMD(p_CellId, cs_WLAN_AP_AssociateCmd));

    f_WLAN_AP_WaitForIndication(t_Watchdog, p_CellId, cr_WLAN_AP_AssociationCmplInd);

    f_WLAN_PreliminaryPass(__FILE__, __LINE__, "WLAN AP successfully associated");

    //Step 2. UE initiates a dynamic selection of ePDG using DNS

    //Step 3. UE initiates Authentication and Authorization (EAP-AKA) using CHAP as defined in 33.402[ZZ] clause 6.5.3

    f_WLAN_IPsecTunnel_WaitForIndication(t_Watchdog, p_CellId, cr_WLAN_IPsecTunnelEstabInd_IKE_SA_INIT);

    //Step 4-9. Check:The UE initiates IPsec tunnel establishment procedure using the IKEv2 protocol as defined in  3GPP TS 33.402[ZZ]; Procedure defined in 23.402[YY] clause 8.2.3 steps 4-9

    f_WLAN_IPsecTunnel_WaitForEAP_AKA_Cmpl(t_Watchdog, p_CellId);

    f_WLAN_IPsecTunnel_WaitForIndication(t_Watchdog, p_CellId, cr_WLAN_IPsecTunnelEstabInd_Cmpl);

    f_WLAN_PreliminaryPass(__FILE__, __LINE__, "IPSec IKEv2 tunnel succesfully established");

    t_Watchdog.stop;

  }    

After change
   function f_WLAN_EstablishIPsecTunnel(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    timer t_Watchdog := tsc_WLANWatchDogDef;

    t_Watchdog.start;

    SYS_IND.send(cas_WLAN_SYSTEM_CMD(p_CellId, cs_WLAN_AP_AssociateCmd));

    f_WLAN_AP_WaitForIndication(t_Watchdog, p_CellId, cr_WLAN_AP_AssociationCmplInd);

    f_WLAN_PreliminaryPass(__FILE__, __LINE__, "WLAN AP successfully associated");

    //Step 2. UE initiates a dynamic selection of ePDG using DNS

    //Step 3. UE initiates Authentication and Authorization (EAP-AKA) using CHAP as defined in 33.402[ZZ] clause 6.5.3

    f_WLAN_IPsecTunnel_WaitForIndication(t_Watchdog, wlan_Cell_NonSpecific, cr_WLAN_IPsecTunnelEstabInd_IKE_SA_INIT);

    //Step 4-9. Check:The UE initiates IPsec tunnel establishment procedure using the IKEv2 protocol as defined in  3GPP TS 33.402[ZZ]; Procedure defined in 23.402[YY] clause 8.2.3 steps 4-9

    f_WLAN_IPsecTunnel_WaitForEAP_AKA_Cmpl(t_Watchdog, wlan_Cell_NonSpecific);

    f_WLAN_IPsecTunnel_WaitForIndication(t_Watchdog, wlan_Cell_NonSpecific, cr_WLAN_IPsecTunnelEstabInd_Cmpl);

    f_WLAN_PreliminaryPass(__FILE__, __LINE__, "IPSec IKEv2 tunnel succesfully established");

    t_Watchdog.stop;

  }    

2.8 f_WLAN_IPCAN_InitialRegistration
	Function name
	f_WLAN_IPCAN_InitialRegistration   

	Reason for change
	WLAN cell needs to be turned on  

	Summary of change
	Added corresponding statement.  

	TTCN module
	IMS/PART3_IPCAN/IPCAN_WLAN.ttcn 

	MCC160 Comment
	Accepted. 


Before change
    function f_WLAN_IPCAN_InitialRegistration(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    // !!!! WLAN ??? any cell power to be set ??? !!!!!

    // Switch on UE

    f_UT_SwitchOnUE(UT);

    // Establish IPsec tunnel

    f_WLAN_EstablishIPsecTunnel(p_CellId);

  }    

After change
   function f_WLAN_IPCAN_InitialRegistration(WLAN_CellId_Type p_CellId) runs on WLAN_PTC

  {

    // !!!! WLAN ??? any cell power to be set ??? !!!!!

    f_WLAN_TurnCellOn(p_CellId);
    // Switch on UE

    f_UT_SwitchOnUE(UT);

    // Establish IPsec tunnel

    f_WLAN_EstablishIPsecTunnel(p_CellId);

  }    

2.9 f_WLAN_IPCAN_Init
	Function name
	f_WLAN_IPCAN_Init   

	Reason for change
	Calling f_IP_Handling_Start before f_WLAN_CellConfig_Def not necessary as it is implicitly done in there

Routing table for IPsec tunnel needs to be included  

	Summary of change
	Removed call to f_IP_Handling_Start
Added call to f_IP_Handling_Start after cell configuration

	TTCN module
	IMS/PART3_IPCAN/IPCAN_WLAN.ttcn

	MCC160 Comment
	Accepted in principle:

The proposed implementation (as well as the original one) fully starts emulation of a (3GPP) PDN including PDN specific address servers (DNS, DHCP, ICMPv6) which shall not be accessed by the UE in case of IMSoWLAN.
( only the IMS server shall be started in addition to the WLAN entities.


Before change
    function f_WLAN_IPCAN_Init(IMS_IPCAN_CO_ORD_PORT p_PortIMS,

                             WLAN_CellId_Type p_CellId,

                             PDN_Index_Type p_PdnIndex) runs on WLAN_PTC

  {

    f_WLAN_Init(WLAN_Only);

    f_IP_Handling_Start(IP, p_PdnIndex, cs_DrbInfo_WLAN(enum2int(p_CellId)));
    f_WLAN_CellConfig_Def(p_CellId, p_PdnIndex);

    f_IMS_IPCAN_SendCoOrdMsg(p_PortIMS, cms_IPCAN_IMS_Response(IPCAN_INIT));

  }    

After change
   function f_WLAN_IPCAN_Init(IMS_IPCAN_CO_ORD_PORT p_PortIMS,

                             WLAN_CellId_Type p_CellId,

                             PDN_Index_Type p_PdnIndex) runs on WLAN_PTC

  {

    f_WLAN_Init(WLAN_Only);

    f_WLAN_CellConfig_Def(p_CellId, p_PdnIndex);

    f_IP_Handling_Start(IP, p_PdnIndex, cs_DrbInfo_IPsecTunnel);
    f_IMS_IPCAN_SendCoOrdMsg(p_PortIMS, cms_IPCAN_IMS_Response(IPCAN_INIT));

  }    

MCC160 Implementation
	  type record IP_StartWLAN_Type { 
    WLAN_AddressInfo_Type AddressInfo,

    IP_DrbInfo_Type     DrbInfo,
    PDN_Index_Type      PdnId optional
  };

	  type record IP_StopWLAN_Type { 

    PDN_Index_Type      PdnId optional

  };

	  type union IP_ConfigReq_Type { 
    IP_Start_Type                       StartIP,

    IP_Stop_Type                        StopIP,           /* @sic R5s131000: Null_Type replaced by IP_Stop_Type sic@ */

    IP_StartWLAN_Type                   StartWLAN,
    IP_StopWLAN_Type                    StopWLAN,

    Null_Type                           GeranCnf,         /* e.g. to confirm GeranTrigger */

    IP_ChangeDrbInfo_Type               ChangeDrbInfo,    /* @sic R5s150648 Change 5.1 sic@ */

    EUTRA_CellId_Type                   ChangeEutraCell,  /* @sic R5s150648 Change 5.1 sic@ */

    IP_SendRouterAdvertisement_Type     SendRouterAdvertisement,

    IP_ExplicitRouting_Type             ExplicitRouting,  /* @sic R5s130681 change 3: handling of RTP data; R5s160131: Start/Stop sic@ */

    Null_Type                           ChangeToRawmode   /* @sic R5s150074 - Additional change sic@ */

  };

	  template (value) IP_ConfigReq_Type cs_IP_StartWlanReq(template (value) WLAN_AddressInfo_Type p_WLAN_AddressInfo,

                                                        template (value) IP_DrbInfo_Type p_DrbInfo,

                                                        template (omit) PDN_Index_Type p_PdnId := omit) :=

  {

    StartWLAN := {

      AddressInfo := p_WLAN_AddressInfo,

      DrbInfo := p_DrbInfo ,

      PdnId := p_PdnId
    }

  };

	  template (value) IP_ConfigReq_Type cs_IP_StopWlanReq(template (omit) PDN_Index_Type p_PdnId := omit) :=

  {


    StopWLAN := {

      PdnId := p_PdnId

    }

  };

	  function f_IP_Handling_StartWLAN(IP_RAT_CTRL_PORT p_Port,

                                   template (value) WLAN_AddressInfo_Type p_WLAN_AddressInfo,

                                   template (value) IP_DrbInfo_Type p_DrbInfo,

                                   template (omit)  PDN_Index_Type p_PdnId := omit)

  {

    fl_IP_Config(p_Port, cs_IP_StartWlanReq(p_WLAN_AddressInfo, p_DrbInfo, p_PdnId));

  }

	  function f_IP_Handling_StopWLAN(IP_RAT_CTRL_PORT p_Port,

                                  template (omit)  PDN_Index_Type p_PdnId := omit)

  {

    fl_IP_Config(p_Port, cs_IP_StopWlanReq(p_PdnId));

  }

	  function f_WLAN_CellConfig_Def(WLAN_CellId_Type p_CellId,

                                 PDN_Index_Type p_PdnIndex := PDN_Internet ,

                                 template (omit) PDN_Index_Type p_PdnToBeStarted := omit) runs on WLAN_PTC

  {

    var template (value) WLAN_AddressInfo_Type v_WLAN_AddressInfo := cs_WLAN_AddressInfo(px_WLAN_UE_AddrIPv4,      /* UE_IPAddressIPv4 */

                                                                                         px_WLAN_UE_AddrIPv6,      /* UE_IPAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* DNS_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* DNS_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* EPDG_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* EmergencyEPDG_ServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6,  /* EmergencyEPDG_ServerAddressIPv6 */

                                                                                         px_WLAN_NetworkAddrIPv4,  /* non3gpp_XCAPServerAddressIPv4 */

                                                                                         px_WLAN_NetworkAddrIPv6); /* non3gpp_XCAP_ServerAddressIPv6 */

    var template (value) WLAN_PDN_AddressInfo_Type v_WLAN_PDN_AddressInfo := fl_WLAN_PDN_IPAddressInfo(p_PdnIndex, valueof(v_WLAN_AddressInfo));

    var template (value) WLAN_EAP_AKA_Cntrl_Type v_EAP_AKA := cs_WLAN_EAP_AKA;

    var WLAN_AP_Parameters_Type v_AP_Parameters := f_WLAN_CellInfo_GetAP_Parameters(p_CellId);

    var WLAN_InitialCellPower_Type v_CellPower := f_WLAN_CellInfo_GetCellPower(p_CellId);

    f_IP_Handling_StartWLAN(IP, v_WLAN_AddressInfo, cs_DrbInfo_WLAN(p_CellId), p_PdnToBeStarted);

    f_WLAN_SS_CommonCellConfig(p_CellId, cas_WLAN_CellConfig_REQ(p_CellId, v_AP_Parameters, v_CellPower));

    f_EPDG_SS_CommonConfig(p_CellId, cas_EPDG_Config_REQ(p_CellId,

                                                         v_WLAN_PDN_AddressInfo,

                                                         cs_IPSec_SA(px_ePDG_Ipsec_IntegrityAlgorithm, px_ePDG_Ipsec_CipheringAlgorithm),

                                                         v_EAP_AKA));

  }

	  function f_WLAN_ReleaseCells(template (omit) PDN_Index_Type p_PdnToBeStopped := omit,

                               WLAN_CellId_Type p_CellId := wlan_Cell_NonSpecific,

                               template (value) WLAN_TimingInfo_Type p_TimingInfo := cs_WLAN_TimingInfoNow, 

                               boolean p_ReleaseEPDGOnly := false) runs on WLAN_PTC 

  {

    if (not p_ReleaseEPDGOnly) { //@sic R5-170924 sic@

      f_WLAN_SS_CommonCellConfig(p_CellId, cas_WLAN_CellRelease_REQ(p_CellId, p_TimingInfo));

    }

    f_EPDG_SS_CommonConfig(p_CellId, cas_EPDG_Release_REQ(p_CellId));

    f_IP_Handling_StopWLAN(IP, p_PdnToBeStopped);

  }

	  function f_WLAN_IPCAN_Init(IMS_IPCAN_CO_ORD_PORT p_PortIMS,

                             WLAN_CellId_Type p_CellId,

                             PDN_Index_Type p_PdnIndex) runs on WLAN_PTC

  {

    var PDN_Index_Type v_PdnToBeStarted := p_PdnIndex; 
    f_WLAN_Init(WLAN_Only);


    f_WLAN_CellConfig_Def(p_CellId, p_PdnIndex, v_PdnToBeStarted);

    f_IMS_IPCAN_SendCoOrdMsg(p_PortIMS, cms_IPCAN_IMS_Response(IPCAN_INIT));

  }

	  function f_WLAN_IPCAN_Release(IMS_IPCAN_CO_ORD_PORT p_PortIMS,

                                PDN_Index_Type p_PdnToBeStopped) runs on WLAN_PTC

  {

    f_WLAN_ReleaseCells(p_PdnToBeStopped);

    f_IMS_IPCAN_SendCoOrdMsg(p_PortIMS);

  }

	  altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;

    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartIP)) {

        v_PdnIndex := v_IP_ConfigReq.StartIP.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartIP.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

        fl_IP_ImsServer_Start(v_PdnIndex);

      } else if (ischosen(v_IP_ConfigReq.StopIP)) {

        if (ischosen(v_IP_ConfigReq.StopIP.PdnId)) {       /* @sic R5s131000: stop particular PDN sic@ */

          i := f_PDN_PdnIndex2Integer(v_IP_ConfigReq.StopIP.PdnId);

          fl_StopPDN(i);

        } else {

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            fl_StopPDN(i);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.StartWLAN)) {

        fl_WLAN_Start(v_IP_ConfigReq.StartWLAN.AddressInfo, v_IP_ConfigReq.StartWLAN.DrbInfo); 
        if (isvalue(v_IP_ConfigReq.StartWLAN.PdnId)) {

          v_PdnIndex := v_IP_ConfigReq.StartWLAN.PdnId;

          v_DrbInfo := valueof(cs_DrbInfo_IPsecTunnel);

          f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

          fl_IP_ImsServer_Start(v_PdnIndex);

        } 
      } else if (ischosen(v_IP_ConfigReq.StopWLAN)) {

        fl_WLAN_Stop();
        if (isvalue(v_IP_ConfigReq.StopWLAN.PdnId)) {

          i := f_PDN_PdnIndex2Integer(v_IP_ConfigReq.StopIP.PdnId);

          fl_StopPDN(i);

        }
      } else if (ischosen(v_IP_ConfigReq.ChangeEutraCell)) {   /* @sic R5s150648 Change 6 sic@ */

        v_EutraCellId := v_IP_ConfigReq.ChangeEutraCell;

        if (f_RoutingTable_ChangeEutraCell(vc_IP_PTC_DRB_RoutingTable, v_EutraCellId)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            f_IP_DrbInfo_ChangeEutraCell(vc_IP_PTC_PDN_MappingTable[i].DefaultDrb, v_EutraCellId);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeDrbInfo)) {   /* @sic R5s150648 Change 6 sic@ */

        v_PdnIndex := v_IP_ConfigReq.ChangeDrbInfo.PdnId;

        v_DrbInfo := v_IP_ConfigReq.ChangeDrbInfo.DrbInfo;

        v_CurrentDrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_PdnIndex);

        if (f_RoutingTable_ChangeDrbInfo(vc_IP_PTC_DRB_RoutingTable, v_CurrentDrbInfo, v_DrbInfo)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        }

      } else if (ischosen(v_IP_ConfigReq.SendRouterAdvertisement)) {

        fl_ICMPv6_SendRouterAdvertisement(v_IP_ConfigReq.SendRouterAdvertisement.PdnId);

      } else if (ischosen(v_IP_ConfigReq.ExplicitRouting)) {

        if (ischosen(v_IP_ConfigReq.ExplicitRouting.Add)) {

          f_IP_PTC_RoutingTable_AddEntry(v_IP_ConfigReq.ExplicitRouting.Add.IpInfo,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DRB,

                                         v_IP_ConfigReq.ExplicitRouting.Add.DataMode);

        } else if (ischosen(v_IP_ConfigReq.ExplicitRouting.Remove)) {  /* @sic R5s160131 sic@ */

          f_IP_PTC_RoutingTable_RemoveEntry(v_IP_ConfigReq.ExplicitRouting.Remove);

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeToRawmode)) {  /* @sic R5s150074 - Additional change sic@ */

        vc_IP_PTC_DRB_RoutingTable := {};

        f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

      }

      p_CtrlPort.send(cs_IP_CommonCnf);

    }

  }


3 Branches executed [Only for Cat B CR]
n/a
4 Execution Log Files [Only for Cat B CR]
4.1 Huawei P10
The Huawei P10 UE passed this test case on R&S CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: R5s170449
· In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5 References [Only for Cat B CR]
	[1]
	R5s170449 



